

#### FEEDBACK - 4

Dr. L.C. Waring of Holywood, Co. Down sent me a tape of his "Tomcat" Editor Assembler which uses the TDL mnemonics - and he has versions for early monitors and Nas-Sys. So far, I've only had a single try at making it work. With my usual skill, I managed to produce all possible error messages in no time at all. If the same thing happens next time I try to use it, I suppose I'll have to read the instructions! It is hardly necessary to add that I suggested he submit it for inclusion in the program library.

Other people have written interesting letters, and I hope they aren't offended by not being mentioned. Now for some of my own work.

#### SORTING IT OUT.

There are many ways of sorting data into order, the fastest I have seen is the "Quicksort". The version below is adapted from a demonstration for the Commodore Pet (whatever that is) printed in the Liverpool Software Gazette (nearly as good as this magazine).

```
10 INPUT "NUMBER OF ITEMS TO SORT";NE
15 IF NE<255 THEN 30
20 PRINT "TOO LARGE." :GOTO 10
30 DIM A(NE),ST(10,2)
35 LE = 1: RI=2
40 FOR I =1 TO NE
50 A(I) = 100 * RND (1): PRINT A (I);
60 NEXT I
70 REM NOW THE SORT ITSELF
80 SP = 1: ST(SP,LE)=1:ST(SP,RI)=NE
85 LR = ST(SP,LE): RR = ST(SP,RI):SP=SP-1
90 I = LR:J=RR:X=A(INT((LR+RR)/2))
100 IF A(I)<XTHEN I=J+1:GOTO 100
110 IF X<A(J)THENJ=J-1:GOTO 110
120 IF I>JTHEN 130
125 W=A(I):A(I)=A(J):A(J)=W:I=I+1:J=J-1
130 IF I<=J THEN 100
135 IF I<RR THEN SP=SP+1:ST(SP,LE)=I:ST(SP,RI)=RR
140 RR=J
145 IF LR<RR THEN 90
150 IF SP<>0 THEN 85
155 PRINT : PRINT
157 REM HERE COME THE RESULTS
160 FOR I=1TO NE
170 PRINT A(I);
180 NEXT I
190 END
```

If you want to sort more than 255 items, the array ST() should be larger. DIM ST(30,2) will allow you to sort much larger arrays, whereupon time taken is extended considerably. It remains much faster than a "Bubblesort" however. The same program will sort strings, if A() is changed to A\$( ), X to X\$ and W to W\$. At 2MHz it takes about 10 seconds to sort fifty words.